

Description

`graph twoway lowess` plots a lowess smooth of *yvar* on *xvar* using `graph twoway line`; see [G-2] [graph twoway line](#).

Quick start

Local linear smooth plot of y versus x using Cleveland's tricube weighting function with a bandwidth of 0.8

```
twoway lowess y x
```

Same as above, overlaid on a scatterplot of y versus x

```
twoway scatter y x || lowess y x
```

Same as above, but draw points with less intense color to make the line more visible

```
twoway scatter y x, mcolor(*.6) || lowess y x
```

Use running-mean smoothing

```
twoway scatter y x || lowess y x, mean
```

Specify a bandwidth of 0.5

```
twoway scatter y x || lowess y x, bwidth(.5)
```

Suppress use of Cleveland's tricube weighting function

```
twoway scatter y x || lowess y x, noweight
```

Menu

Graphics > Two-way graph (scatter, line, etc.)

Syntax

```
twoway lowess yvar xvar [if] [in] [, options]
```

<i>options</i>	Description
<code>bwidth(#)</code>	smoothing parameter
<code>mean</code>	use running-mean smoothing
<code>noweight</code>	use unweighted smoothing
<code>logit</code>	transform the smooth to logits
<code>adjust</code>	adjust smooth's mean to equal <i>yvar</i> 's mean
<i>cline_options</i>	change look of the line
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

Options

`bwidth(#)` specifies the bandwidth. `bwidth(.8)` is the default. Centered subsets of $N \times \text{bwidth}()$ observations, N = number of observations, are used for calculating smoothed values for each point in the data except for endpoints, where smaller, uncentered subsets are used. The greater the `bwidth()`, the greater the smoothing.

`mean` specifies running-mean smoothing; the default is running-line least-squares smoothing.

`noweight` prevents the use of Cleveland's (1979) tricube weighting function; the default is to use the weighting function.

`logit` transforms the smoothed *yvar* into logits.

`adjust` adjusts by multiplication the mean of the smoothed *yvar* to equal the mean of *yvar*. This is useful when smoothing binary (0/1) data.

cline_options specify how the lowess line is rendered and its appearance; see [G-3] *cline_options*.

axis_choice_options associate the plot with a particular *y* or *x* axis on the graph; see [G-3] *axis_choice_options*.

twoway_options are a set of common options supported by all *twoway* graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] *twoway_options*.

Remarks and examples

`graph twoway lowess yvar xvar` uses the `lowess` command—see [R] [lowess](#)—to obtain a local linear smooth of *yvar* on *xvar* and uses `graph twoway line` to plot the result.

Remarks are presented under the following headings:


Typical use
Use with by()

Typical use

The local linear smooth is often graphed on top of the data, possibly with other regression lines:

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)

. twoway scatter mpg weight      ||
      lfit   mpg weight        ||
      lowess mpg weight
```




Use with by()

graph twoway lowess may be used with by():

```
. use https://www.stata-press.com/data/r19/auto, clear
(1978 automobile data)

. twoway scatter mpg weight ||
    lfit   mpg weight ||
    lowess mpg weight ||, by(foreign)
```



References

- Cleveland, W. S. 1979. Robust locally weighted regression and smoothing scatterplots. *Journal of the American Statistical Association* 74: 829–836. <https://doi.org/10.2307/2286407>.
- Cox, N. J. 2005. Speaking Stata: Smoothing in various directions. *Stata Journal* 5: 574–593.
- . 2010. Software Updates: Speaking Stata: Smoothing in various directions. *Stata Journal* 10: 164.
- Royston, P., and N. J. Cox. 2005. A multivariable scatterplot smoother. *Stata Journal* 5: 405–412.

Also see

[R] [lowess](#) — Lowess smoothing

[G-2] [graph twoway mspline](#) — Two-way median-spline plots

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

